' RRSP Calculator - © 1993-2002 Aiman Hanna

' This program explores more of the repetition structures,

' in specific, the For/Next' repetition structure.

' The program also introduces List Boxes as well as strings

' can be formatted.

' Key Points:

'       1) For/Next repetition structure

'       2) List Boxes

'       3) Format$

Option Explicit

Private Sub cmdShowRRSP_Click()

' This function will calculate the RRSP investment over a number of years.

' The function starts with a principle amount, interest rate. The principle amount

' will increase every year by the interest rate. The formula used is as follows:

' this year's amount = previous year's amount + interest accumulated. For the

' first year, the principle is used as previous year's amount.

    Dim amount As Single, rate As Single, interest As Single

    Dim years As Integer, counter As Integer

    Dim ans As Integer

    Dim str As String

    amount = txtAmount.Text

    years = txtYears.Text

    rate = txtRate.Text

    ' clear the list

    lstDetails.Clear

    If (years < 1) Or (years > 25) Then

        Beep

        ans = MsgBox("Enter a valid number of years (between 1 & 25)", vbOKOnly, "Invalid Number of Years")

    Else    ' number of years is good

        For counter = 1 To years    ' This is the same as: For counter = 1 To years Step 1

            interest = amount * (rate / 100)    ' divide by 100 since rate is a % value

            amount = amount + interest

            str = "Year # " & Format$(counter, "@@@") & ":" & vbTab & "Balance is: " _

                  & Format$(Format$(amount, "Currency"), String$(15, "@"))

            lstDetails.AddItem (str)

        Next counter

    End If

End Sub

Private Sub cmdExit_Click()

    End

End Sub

